
Basic Super-Oric

Mise au point Software pour Cartouches Super-Oric  (16)
Les commandes HIRES (suite et fin)

    par André C.

CONSIDÉRATIONS GÉNÉRALES

Au cours de ma série d'articles, j'ai utilisé plusieurs
commandes HIRES. Vous avez pu vous avez pu
voir qu'à part la commande FILL, elles se com-
portent toutes comme avec l'Oric classique. Petite
correction: la commande CHAR n'est plus opéra-
tionnelle. En effet, puisqu'en mode HIRES nous
disposons toujours de l'écran texte (modes LORES
0 ou LORES 1), la commande CHAR est devenue
inutile (nous verrons un exemple). Fabrice sem-
ble donc l'avoir laissée tomber. En fait, cette com-
mande est toujours là (et génère même des messa-
ges d'erreurs lorsque la syntaxe est incorrecte),
mais elle ne fonctionne plus.
Pour toutes les commandes HIRES restantes, soit
(dans l'ordre alphabétique) CIRCLE, CURMOV,
CURSET, DRAW, FILL et PATTERN, il suffit de
tenir compte des nouvelles dimensions de l'écran
HIRES (256x224 pixels) et/ou de l'augmentation
du nombre de couleurs (jusqu'à 256) et aussi que
d'une différence dans le comportement de la va-
leur 2 du paramètre fb (foreground/background ou
encre/papier).

LES COORDONNÉES X ET Y
Elles acceptent maintenant des valeurs comprises
entre 0 et 255 (de gauche à droite) pour X et 0 et
223 (de haut en bas) pour Y. Le reste fonctionne
comme avec l'Oric classique. Selon les comman-

des, les valeurs de X et Y peuvent être absolues
(respectivement de 0 à 255 et de 0 à 223) ou rela-
tives (respectivement de -255 à + 255 et de -223 à
+223). Pas de difficulté donc, mais il faut éviter
de sortir de l'écran, une fantaisie qui est immédia-
tement sanctionnée par un message d'erreur. Les
tracés HIRES se font dans la couleur de l'encre
courante (une parmi 256) sélectionnée grâce à la
commande INK (voir le dernier article).

LE PARAMÈTRE FB

Comme dans l'Oric classique, c'est la valeur 1 du
paramètre fb qui permet de tracer dans la couleur
courante de l'encre. Par défaut (au boot), cette va-
leur courante est la couleur n°1, prédéfinie en
blanc. Par la suite, après définition d'autres cou-
leurs avec la commande DEF INK, la valeur cou-
rante est sélectionnée par la commande INK.
Le paramètre fb=0 (tracé de la couleur du fond,
avec surcharge des pixels déjà éventuellement
dessinés dans une autre couleur) existe toujours,
car très pratique puisqu'il simplifie la program-
mation qui aurait consisté à indiquer l'encre n°0.
Par défaut (au boot), cette couleur n°0 est
prédéfinie en noir. Par la suite, on peut définit n'im-
porte quelle autre couleur de fond avec la com-
mande DEF INK 0, valeur. Ce changement affec-
tera non seulement le fond, mais aussi tout élé-
ment HIRES tracé dans la couleur n°0 précédente.
Je rappelle que ce paramètre fb=0 est très prati-
que pour effacer un élément HIRES existant.
Le paramètre fb=3 (ne rien changer à la couleur
des pixels sous-jacents) existe toujours, inchangé.
C'est vraiment le plus simple, dommage qu'on ne
puisse pas s'en tenir à lui...
Quant au paramètre 2, à propos duquel Fabrice écrit:
"CURSET x,y,2 fait une inversion de la couleur du
point : Un point de la couleur d'encre prendra la
couleur du papier et inversement (plus exactement,
un OU exclusif est effectué entre le numéro de la
couleur du point et le numéro de la couleur d'encre
: on obtient donc une tierce couleur si la couleur
d'encre a été changée, mais on garde le comporte-
ment Oric où la couleur d'un point est restituée après
deux inversions successives)." Je cite, car cette his-
toire n'est pas facile à comprendre et à dire vrai, je
n'ai pas compris. Voici un exemple, mais je recon-
nais qu'il faudrait faire mieux.

Comparaison Atmos versus Super-Oric
Bien que les syntaxes soient identiques, les résul-
tats obtenus diffèrent considérablement. Dans le
programme suivant, il s'agit de tracer 4 lignes ho-
rizontales de couleurs différentes et de paramè-
tres fb différents (de 0 à 3), puis 4 lignes vertica-
les avec les mêmes variations. Il sera intéressant
d'observer non seulement ces lignes, mais aussi
leurs points d'intersection.
Version Atmos
Les événements évoluant très vite avec l'Atmos,
j'ai été obligé d'insérer quelques WAIT. Notez par
ailleurs le recours à la commande CHAR pour
identifier les lignes et leur paramètre fb:
100 HIRES:PAPER 7:' Blanc
102 W=100:' Délai
105 ' Lignes horizontales
110 INK 0:CURSET 50,100,3:CHAR #30,0,1
112 CURMOV 6,4,3:DRAW 100,0,0
115 ' Noir, fb=Paper
117 WAIT W
120 INK 1:CURSET 50,110,3:CHAR #31,0,1
122 CURMOV 6,4,3:DRAW 100,0,1
125 ' Rouge, fb=Ink
127 WAIT W
130 INK 2:CURSET 50,120,3:CHAR #32,0,1
132 CURMOV 6,4,3:DRAW 100,0,2
135 ' Vert, fb=inv Paper/Ink
137 WAIT W
140 INK 3:CURSET 50,130,3:CHAR #33,0,1
142 CURMOV 6,4,3:DRAW 100,0,3
145 ' Jaune, fb=ne change pas pixel
existant
146 ' Lignes verticales

147 WAIT W
150 INK 0:CURSET 100,50,3:CHAR #30,0,1
152 CURMOV 2,10,3:DRAW 0,100,0
155 ' Noir, Paper
157 WAIT W
160 INK 1:CURSET 110,50,3:CHAR #31,0,1
162 CURMOV 2,10,3:DRAW 0,100,1
165 ' Rouge, Ink
167 WAIT W
170 INK 2:CURSET 120,50,3:CHAR #32,0,1
172 CURMOV 2,10,3:DRAW 0,100,2
175 ' Vert, Couleurs inverses
177 WAIT W
180 INK 3:CURSET 130,50,3:CHAR #33,0,1
182 CURMOV 2,10,3:DRAW 0,100,3
185 ' Jaune, fb=ne change pas pixel
existant
187 END

La figure ci-dessus témoigne de l'évolution de
l'écran HIRES et le gros plan montre les points
d'intersection. On y voit:
1) Que fb=0 ne trace rien lorsque l'écran est vierge
(blanc), mais trace des points blancs lorsque la li-
gne surcharge des éléments existants.
2) Que fb=1 impose une ligne continue blanche
quel que soit ce qu'il y a dessous, mais la ligne
tracée peut être surchargée à son tour.
3) Que fb=2 inverse la couleur des pixels sous-
jacents pour tracer sa ligne. Là où les pixels étaient
de couleur PAPER (ici blanc) il écrit dans la cou-
leur INK (par exemple vert) et là ou les pixels
étaient de couleur INK (par exemple INK), il écrit
dans la couleur PAPER (ici blanc). On voit nette-
ment ce blanc aux points d'intersection des lignes.
4) Que fb=3 ne fait rien (que c'est reposant de ne
rien faire).

Cette figure montre les 8 étapes du programme pour Atmos, plus un gros plan des intersections

Mais nous voyons aussi que tout changement de
la couleur INK altère l'ensemble des tracés HIRES.
Version Super-Oric
Le programme a été adapté au Super-Oric pour
remplir le même cahier des charges. On voit tout
d'abord que j'ai redéfini les 8 premières couleurs
du Super-Oric avec les 8 couleurs de l'Oric classi-
que, en respectant leur ordre de 0 à 7: En consé-
quence l'écran sera noir. Ensuite, j'ai amélioré la
lisibilité des résultats en traçant, devant chaque
100 HIRES:DEF INK 0,0,#1F,#3E0,#3FF,#7C00,#7C1F,#7FE0,#7FFF
101 ' Noir, Rouge, Vert, Jaune, Bleu, magenta, Cyan, Blanc
102 ' Le papier est Noir
105 ' Lignes horizontales
110 INK 1:CURSET 52,100,3:DRAW 10,0,1:DRAW 180,0,0
115 ' Rouge, fb=Paper
120 INK 2:CURSET 52,116,3:DRAW 10,0,1:DRAW 180,0,1
125 ' Vert, fb=Ink
130 INK 3:CURSET 52,132,3:DRAW 10,0,1:DRAW 180,0,2
135 ' Jaune, fb=n°CoulPixel XOR n°CoulInk
140 INK 4:CURSET 52,148,3:DRAW 10,0,1:DRAW 180,0,3
145 ' Bleu, fb=ne change pas pixels existants
146 ' Lignes verticales
150 INK 4:CURSET 100,60,3:DRAW 0,10,1:DRAW 0,100,0
155 ' Bleu, fb=Paper
160 INK 5:CURSET 140,60,3:DRAW 0,10,1:DRAW 0,100,1
165 ' Magenta, fb=Ink
170 INK 6:CURSET 180,60,3:DRAW 0,10,1:DRAW 0,100,2
175 ' Cyan, fb=n°CoulPixel XOR n°CoulInk
180 INK 7:CURSET 220,60,3:DRAW 0,10,1:DRAW 0,100,3
185 ' Blanc, fb=ne change pas pixels existants
220 PLOT 2,12,"fb=0":PLOT 10,6,"fb=0"
230 PLOT 2,14,"fb=1":PLOT 15,6,"fb=1"
240 PLOT 2,16,"fb=2":PLOT 20,6,"fb=2"
250 PLOT 2,18,"fb=3":PLOT 25,6,"fb=3"

ligne, un petit segment de couleur identique, mais
avec fb=1 au lieu de fb variable de 0 à 3. Enfin, le
recours à la commande CHAR a été remplacé par
des PLOT dans l'écran TEXT qui coexiste avec
l'écran HIRES. Voici le listing et le résultat qui
tient en une seule recopie d'écran (plus un gros
plan sur les intersections), car avec le Super-Oric,
la commande INK n'altère pas les éléments HIRES
déjà tracés. On a donc directement un dessin mul-
ticolore, qui n'aurait été possible dans l'Oric clas-
sique qu'avec l'insertion d'attributs de couleurs.

Cette figure montre les 3 variantes du programme pour Super-Oric avec les gros plans correspondants

Les conclusions pour le Super-Oric et la compa-
raison avec l'Atmos sont les suivantes
1) fb=0 ne trace rien lorsque l'écran est vierge (ici
noir), mais trace des points noirs lorsque la ligne
surcharge des éléments existants. Donc compor-
tement classique.
2) fb=1 impose une ligne continue de la couleur
INK courante, quel que soit ce qu'il y a dessous,
mais la ligne tracée peut à être surchargée à son
tour. Donc comportement classique.
3) fb=2 semble faire la même chose que fb=0, c'est

Après avoir défini les encres n°0 (rouge) et n°1
(cyan), ce programme trace les lignes verticales
puis horizontales d'une grille rouge sur fond cyan.
Rien à dire de plus. La recopie d'écran ci-dessous
montre le résultat:

à dire tracer avec la couleur n°0. Donc, à première
vue, comportement différent de l'Oric classique.
4) fb=3 ne fait rien. Donc comportement classique.
Contrairement à l'Oric classique, tout changement
de la couleur INK n'altère que les prochains tra-
cés HIRES et non l'ensemble des tracés.
Cette histoire de fb=2 n'étant pas claire, je tente
donc d'utiliser des couleurs non classiques:
l'orange #21F et son inverse exact le bleu2 #7DE0.
Voici les lignes modifiées:

100 HIRES:DEF INK 0,0,#1F,#3E0,#21F,#7C00,#7C1F,#7DE0,#7FFF
101 ' Noir, Rouge, Vert, Orange, Bleu, magenta, Bleu2, Blanc
135 ' Orange, fb=n°CoulPixel XOR n°CoulInk
175 ' Bleu2, fb=n°CoulPixel XOR n°CoulInk

Et le résultat: Nous ne sommes pas plus avancés.
Avec fb=2, le trait Orange a encore été tracé en
noir et le trait bleu2 aussi. On voit même le croise-
ment avec le trait vert.

Pour voir ce que ça change, je tente le coup de mettre
le fond en Orange de tracer avec fb=2 une première
ligne (horizontale) bleue2 et une seconde (verticale)
orange. Les lignes suivantes sont donc modifiées:

100 HIRES:DEF INK 0,#21F,#1F,#3E0,#7DE0,#7C00,#7C1F,#21F,#7FFF
101 ' Orange, Rouge, Vert, Bleu2, Bleu, Magenta, Orange, Blanc
102 ' Le papier est Orange
135 ' Bleu2, fb=n°CoulPixel XOR n°CoulInk
175 ' Orange, fb=n°CoulPixel XOR n°CoulInk

Et le résultat est toujours aussi surprenant, puis-
que avec fb=2, on a toujours nos deux lignes tra-
cées en noir (que cette fois on distingue du fond,
il ne s'agit donc pas de la couleur n°0, mais d'un
noir qui a été défini non pas volontairement avec
DEF INK, mais par la console Snes. Toutefois, la
double inversion s'est produite, car à l'intersection
des deux lignes fb=2 on distingue un pixel Orange.
J'ai tenté de comprendre en triturant avec XOR
les n° des couleurs (et même leur valeur de défini-
tion), mais ne suis arrivé à rien. Donc, pour l'ins-
tant, le paramètre fb=2 n'est pas utilisable (en tout
cas par moi), mais cela ne me semble pas bien
grave.

CURSET & DRAW
Nous avons déjà souvent utilisé ces commandes
au cours de cette série d'article. C'est vraiment pour
le plaisir que je vous donne l'exemple suivant:
5 HIRES:DEF INK 0,31*1056,31
10 FOR X=0 TO 255 STEP 10
20 CURSET X,0,0
30 DRAW 0,223,1
40 NEXT
50 FOR Y=0 TO 223 STEP 10
60 CURSET 0,Y,0
70 DRAW 255,0,1
80 NEXT

PATTERN & CURMOV
Et enfin la commande PATTERN, qui est absolu-
ment identique à celle de l'Oric classique. La syn-
taxe est simple: PATTERN valeur (comprise en-
tre 0 et 255). Tiens, il y a 256 valeurs possibles et
256 colonnes dans l'écran HIRES du Super-Oric.
Nous allons donc tracer 256 lignes verticales cha-
cune avec un pattern différent:

10 REM.....PATTERN.....
20 HIRES
30 DEF INK 0,31,0
25 CURSET 0,0,3
30 FOR X=0 TO 255
40 PATTERN X
50 DRAW 0,223,1
60 CURMOV 1,-223,0
70 NEXT

Comme d'habitude, après chaque DRAW, le cur-
seur HIRES reste positionné à la fin de la ligne
tracée. La commande CURMOV le repositionne
donc au début de la ligne suivante: avec x=1 on
passe à la colonne suivante et avec y=-223 on re-
monte en haut de l'écran. C'est une alternative à
l'utilisation de CURSET comme dans le pro-
gramme précédent. J'ai été surpris par l'erreur ob-
tenue à la fin du remplissage de l'écran. Mais j'ai
vite compris que passer à la colonne suivante
quand on a déjà atteint la dernière, constitue une
sortie d'écran. Voici donc la recopie d'écran avec
son message d'erreur et ci-dessous un agrandisse-
ment du haut de l�écran. La première colonne se
réduit à une simple ligne rouge (PATTERN 0, rien
n'est tracé, le motif étant 00000000). Sur la se-
conde on voit un pixel noir tous les 256 pixels
correspondant à la valeur 1 (motif 00000001). La
colonne suivante montre le dessin de la valeur
00000010 (2 en décimal), la suivante 00000011
(3 en décimal) etc. Et la dernière colonne, à droite
de l'écran est une ligne continue correspondant à
la répétition du motif 11111111.

Pour obtenir de petits pointillés réguliers il faut
avoir le motif 01010101 soit #55 ou 85 en déci-
mal. Pour un pointillé régulier plus gros, le motif
00110011 conviendra (#33 ou 51 en décimal). Les
S.O.S. s'écrivent avec 101101011 ou 11011110 etc.
Rien de neuf sous le soleil donc!

EN CONCLUSION

Sauf l'absence de la commande CHAR, aisément
remplaçable par la commande PLOT et le com-
portement bizarre du paramètre fb=2, les comman-
des HIRES du Super-Oric ont un comportement
proche, voire identique à leurs homologues de
l'Oric classique. Si on tient compte du fait que
l'écran HIRES est un peu plus grand et du fait que
la commande INK permet de tracer dans 256 cou-
leurs différentes, il n'y a aucune difficulté dans l'uti-
lisation des commandes HIRES du Super-Oric.

Bon amusement donc...

 

On ne dit pas « Faire des 
vendanges », mais� 

Voyons� Je sais :  
« Péter comme un Dieu ! » 

